
Celsius en Fahrenheit

Omrekenformules

Uitgangspunt

celsius * 1,8 = fahrenheit - 32

noForth heeft geen floating point

Daarom:

celsius * 9/5 = fahrenheit - 32

Hoe kom je nu aan de omrekenformules?

Linkerkant en rechterkant van 'Uitgangspunt' zijn aan elkaar
gelijk. Dat blijft zo als je bij beide kanten 32 optelt of als je
beide kanten met 5/9 vermenigvuldigt.

Plus 32 aan beide kanten

celsius * 9/5      = fahrenheit - 32
celsius * 9/5 + 32 = fahrenheit

: >F ( cels -- fahr ) 9 * 5 / 32 + ;

Maal 5/9 aan beide kanten

celsius * 9/5 =   fahrenheit - 32
celsius       = ( fahrenheit - 32 ) * 5/9

: >C ( fahr -- cels ) 32 - 5 * 9 / ;

Inschalen

Om te werken met tienden van graden als eenheid is het
voldoende om 32 in 320 te veranderen.

celsius * 9/5 = fahrenheit - 320

: >C10 ( fahr*10 -- cels*10 ) 9 * 5 / 320 + ;
: >F10 ( cels*10 -- fahr*10 ) 320 - 5 * 9 / ;

Star-slash

*/ is een standaardwoord. Het heeft als extra dat het
tussenresultaat (x*y) een double number is. Daardoor heb je
geen last van de overflow die kan ontstaan bij de gewone
vermenigvuldiging. 
Voor >F en >C dreigt dat gevaar alleen als je nog kleinere
eenheden kiest.

: */ ( x y x -- x*y/z ) >r m* r> fm/mod nip ;



Nauwkeurigheid

  discussie onderwerpen 

Bij een deling in forth wordt er afgerond. Het is daarom
onverstandig om eerst te delen en daarna te vermenig-vuldigen,
zoals in

: >F ( cels -- fahr ) 5 / 9 * 32 + ;

Bij de deling door 5 kan de afronding hoogstens 1 afwijken.
Door daarna met 9 te vermenigvuldigen kan de afwijking tot 9
oplopen!

Afronding

Als een resultaat is afgerond op bijvoorbeeld 24 betekent dat in
het algemeen dat de werkelijke waarde tussen 23,5 en 24,5 zal
liggen. 
Bij een deling in forth betekent een resultaat van 24 dat de
werkelijke waarde tussen 24 en 25 ligt, want er wordt altijd naar
beneden afgerond. 
Je kunt in forth een deling x/y op de klassieke manier afronden
door x vooraf te verhogen met de helft van y.

x/y   -->   ( x + y/2 ) / y

Dat kan in ons geval eenvoudig met de hand:

: >F ( cels -- fahr ) 9 *   2 + 5 /   32 + ;
: >C ( fahr -- cels ) 32 -   5 *   4 + 9 / ;

Bij modulo-deling (floored division) gaat dit ook goed voor
negatieve temperaturen. NoForth heeft modulo-deling. 
Bij symmetrische deling wordt het ingewikkelder zodra er
negatieve getallen bij betrokken zijn.

Negatieve getallen afronden

Een eenvoudige manier van afronden bij delingen waar
negatieve getallen bij kunnen optreden is: 
- bepaal vooraf het teken van de uitkomst, 
- maak daarna alles positief, 
- voeg na de deling het teken weer toe. 
Dit werkt voor beide delingtypes.

: >F ( cels -- fahr )
    dup abs 9 * 2 + 5 / swap ?negate   32 + ;
: >C ( fahr -- cels )
    32 -   dup abs 5 * 4 + 9 / swap ?negate ;

?NEGATE is geen standaardwoord, maar is wel vrij algemeen.
Het bestaat ook in noForth.

: ?NEGATE ( x1 y -- x2 ) 0< if negate then ;

Tenslotte

Door in te schalen naar kleinere eenheden kun je de
afrondingsfouten zo klein maken dat ze niet meer relevant zijn
voor het doel dat je wilt bereiken.

