

Een groot assembler voorbeeld, een 200 Hz interrupt die een SPI display bestuurt.

```
routine >SPI      ( -- adr ) \ Send byte in SUN using SPI
  8 zz mov
  begin,
    sun sun .b add \ Get highest bit
    cs? if,        \ High?
      #2 29 & .b bis \ P2OUT  P2.1  Yes
    else,
      #2 29 & .b bic \ P2OUT  P2.1  No
    then,
      #4 29 & .b bis \ P2OUT  P2.2  Clock
      #4 29 & .b bic \ P2OUT
      #-1 zz add    \ Decrease bit counter
  =? until,        \ Zero?
  rp )+ pc mov     \ Ready
end-code
```

\ Display data from the array digits on 4-digit tube display
\ This interrupt has to be triggered 100 to 200 times each second!

```
routine TUBE      ( -- adr )
  sun push        \ Save Forth registers
  day push
  adr segm & day mov \ Contents of segm
  digits # yy mov  \ Addr of number array
  4 # xx cmp       \ All numbers done?
  =? if,          \ Yes,
    #0 xx mov      \ restart digit pointer
    #8 day mov     \ and display selector
  then,
    xx yy add      \ Yes, add pointer
    yy ) sun .b mov \ Read bitmap
    >spi # call     \ Send bitmap
    day sun .b mov  \ Get segment
    >spi # call     \ Select active display
    #1 29 & .b bis  \ P2OUT  P2.0  Display one digit
    #1 29 & .b bic  \ P2OUT
    day .b rrc     \ To next digit
    #1 xx add      \ To next digit
    #1 21 & xor>    \ P1OUT  Toggle P1.0 for debug
    day adr segm & mov \ Save display selector
    rp )+ day mov   \ Restore Forth registers
    rp )+ sun mov
    reti
end-code
```